NASA TECHNICAL MEMORANDUM

NASA TM X-1762

NASA TM X-1762

APPLICATION OF A MODIFIED FAST FOURIER TRANSFORM TO CALCULATE HUMAN OPERATOR DESCRIBING FUNCTIONS

by Richard S. Shirley

Electronics Research Center

Cambridge, Mass.

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • MARCH 1969

NASA TM X-1762

APPLICATION OF A MODIFIED FAST FOURIER TRANSFORM TO CALCULATE HUMAN OPERATOR DESCRIBING FUNCTIONS

By Richard S. Shirley

Electronics Research Center
Cambridge, Mass.

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

For sale by the Clearinghouse for Federal Scientific and Technical Information
Springfield, Virginia 22151 – CFSTI price $3.00

APPLICATION OF A MODIFIED FAST FOURIER TRANSFORM TO CALCULATE HUMAN OPERATOR DESCRIBING FUNCTIONS

By Richard S. Shirley
Electronics Research Center

SUMMARY

A version of the fast Fourier transform (FFT) is used in a hybrid computer program to permit processing of tracking data to yield the human operator's describing function almost immediately after the period of data-taking. The use of the FFT allows the final calculation time required to process 216 seconds of tracking data to be reduced to 3 seconds from the 10 minutes previously required on the same computer. The algorithm used permits the bulk of the analysis of the data to be performed while the data are being taken, and does not require all the data to be present in core before processing begins.

TABLE OF SYMBOLS

a — the index of summation for the additive portion of the FFT

A_k — a constant which weights the sinusoids composing the system input, see Table I

A_{xk} — the real part of the truncated Fourier transform of x(t) at the frequency ω_k, given by

$$\int_0^T x(t) \cos(\omega_k t)\, dt$$

B_{xk} — the imaginary part of the truncated Fourier transform of x(t) at the frequency ω_k, given by

$$\int_0^T x(t) \sin(\omega_k t)\, dt$$

c — a subscript referring to the output of the human operator at the control stick (see Figures 1 and 4)

$c(n\Delta t)$ — the data samples taken at the human operator's output

D_k — an integer devisable by 4 used to determine the input frequencies

e — a subscript referring to the input to the human operator at the oscilloscope (see Figures 1 and 4)

$e(n\Delta t)$ — the data samples taken at the human operator input

FFT — Fast Fourier Transform

$F_x(\omega_k)$ — the truncated Fourier transform of x(t) at the frequency ω_k, given by

$$\int_0^T x(t)e^{-j\omega t}\,dt$$

h — a subscript used to denote frequencies between the input frequencies, equal to 1, 2, 3,...

i — a subscript referring to the system input (see Figures 1 and 4)

$i(n\Delta t)$ — the system input equals

$$\sum_{k=1}^{14} A_k \sin\,(\omega_k n\Delta t)$$

j — the square root of -1.

k — a subscript used to denote the input frequencies, equal to 1, 2, 3,...,14

$m(n\Delta t)$ — the data samples taken at the system output

n — the index of summation for the multiplicative portion of the FFT

N — the number of data samples taken, 10,800

T — the period of data-taking, equal to 216 seconds

$Y_c(\omega)$ — the dynamics of the controlled element (see Figures 1 and 4)

$Y_p(\omega)$ — the linear portion of the quasi-linear describing function

β_k — $(N/D_k) - 1$

γ_k — $(D_k/4) - 1$

Δt — the time increment between interrupts, and hence the time between data samples, equals .02 sec

ω_k — the frequencies of the sinusoids comprising the system input, see Table I

ω_h — frequencies between the ω_k's

$\Phi_{ic}(\omega)$ — the cross power spectral density between the human operator's output and the system input

$\Phi_{ie}(\omega)$ — the cross power spectral density between the human operator's input and the system input

$\Phi_{nn}(\omega)$ — the continuous power spectral density of the human operator's remnant

INTRODUCTION

Only recently have dynamic models of the human operator been used effectively in the design of man-vehicle systems. This is due partially to a lack of understanding of the human operator and also to the difficulty and expense of experimentally determining values for the various parameters of existing models. Improvements in computers and computational techniques are overcoming these difficulties, and already it is possible to bring about significant improvements in a man-vehicle system through the use of pilot models in preliminary design (refs. 1 and 2). This paper describes a computational technique which reduces greatly the cost of obtaining values permitting the use of a current pilot model, i.e., the quasi-linear describing function.

One way to characterize the behavior of a human operator in a continuous tracking task is by a quasi-linear describing function, which consists of a linear describing function and a remnant. The linear describing function is the average frequency response of the human operator, i.e., his amplitude ratio and phase as a function of frequency. The remnant, characterized by a continuous power spectral density, is that portion of the human operator's output which is not linearly correlated with his input. The total output of the human operator is the sum of the remnant and the output of the linear describing function* (see Figure 1).

---

*Examples of human operator describing functions are shown in Figures 2 and 3.



Figure 1.- Block diagram of the human operator in a compensatory tracking system.



Figure 2.- $Y_p(\omega)$ measured for $Y_c(s) = 1/s$

A direct way to measure describing functions in the laboratory involves the use of a hybrid computer and the method of Fourier coefficients. The method of Fourier coefficients has been extensively investigated and is described in detail (ref. 3). It will be briefly outlined here for completeness. The human operator is placed in a control loop, possibly as shown in Figures 1 and 4. The system input, a sum of sinusoids of known amplitude, phase, and frequency is updated every Δt seconds; simultaneously, data are taken at the human operator's input and output. At the end of T seconds, the sampled values of the human operator's input and output are processed as follows:

$$A_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{N} c(n\Delta t)\ \cos\ (\omega_k n\Delta t) \tag{1}$$

$$B_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{N} c(n\Delta t)\ \sin\ (\omega_k n\Delta t) \tag{2}$$

$$A_{ek} = \frac{\Delta t}{T}\sum_{n=1}^{N} e(n\Delta t)\ \cos\ (\omega_k n\Delta t) \tag{3}$$

$$B_{ek} = \frac{\Delta t}{T}\sum_{n=1}^{N} e(n\Delta t)\ \sin\ (\omega_k n\Delta t) \tag{4}$$

$$F_c(\omega_k) = A_{ck} - jB_{ck} \tag{5}$$

$$F_e(\omega_k) = A_{ek} - jB_{ek} \tag{6}$$

$$Y_p(\omega_k) = \frac{F_c(\omega_k)}{F_e(\omega_k)} \tag{7}$$

$$\underline{/Y_p(\omega_k)} = -\tan^{-1}\frac{B_{ck}}{A_{ck}} + \tan^{-1}\frac{B_{ek}}{A_{ek}} \tag{8}$$



Figure 3.- $Y_p(\omega)$ measured for $Y_c(s) = 1/s^2$



Figure 4.- Flow diagram of the human operator in a compensatory tracking system.

$$\left|Y_p(\omega_k)\right| = \left[\frac{A_{ck}^2 + B_{ck}^2}{A_{ek}^2 + B_{ek}^2}\right]^{1/2} \tag{9}$$

$$\Phi_{cc}(\omega_h) = \frac{1}{2\pi T}\left|F_c(\omega_h)\right|^2 \tag{10}$$

$$\Phi_{nn}(\omega_h) = \Phi_{cc}(\omega_h)\left|1 + Y_p Y_c(\omega_h)\right|^2 \tag{11}$$

where the ω_k's are the input frequencies, and the ω_h's lie between the ω_k's.

This paper describes how a version of the fast Fourier transform (FFT) is used to compute human operator describing functions, or more specifically, how a version of the FFT is used to solve Eqs. (1) through (4), while the data samples, $c(n\Delta t)$ and $e(n\Delta t)$, are being taken. The FFT is an algorithm which greatly reduces the time required to calculate the truncated Fourier transform, or periodogram, of a sampled time signal. The savings are obtained by replacing calculations which involve trigonometric functions or multiplications with simple additions. The replacement is accomplished by taking advantage of the symmetries of the sine and cosine functions, and by further taking advantage of relationships between the frequencies at which the Fourier analysis is performed.

THE VERSION OF THE FFT USED

The version of the FFT used takes advantage only of the symmetries of the sine and cosine functions. It does not take advantage of the relationships among the frequencies at which the Fourier analysis is performed. By not using the complete version of the FFT, it becomes possible to perform the bulk of the data-processing during the Δt seconds between interrupts while the experiment is still in process. The requirement that the data be in core before processing, or even that the data fit in core, is avoided. The following derivation of the algorithm used will make this point clearer. It should be noted that before the FFT was used it was not possible to perform the calculations between interrupts because the computation time required was over two and a half times greater than that which was available.

It is desired to evaluate Eqs. (1) through (4) using a digital computer. In order to permit the use of the FFT, the input frequencies, ω_k, will be restricted to

$$\omega_k = \frac{2\pi}{D_k \Delta t}$$

where the D_k are chosen from 4, 8, 12, 16, etc. The method of Fourier coefficients further requires that the ratio N/D_k be an integer (where N is the number of data samples taken at intervals Δt). The derivation for A_{ek} and B_{ek} is identical to the derivation which follows for A_{ck} and B_{ck}.

Using the identities

$$\sin(\theta + 2\pi) = \sin\theta, \text{ and}$$

$$\cos(\theta + 2\pi) = \cos\theta$$

or

$$\left.\begin{aligned} \sin(\omega_k \Delta t) &= \sin\left[(aD_k + 1)\omega_k \Delta t\right] \\ \cos(\omega_k \Delta t) &= \cos\left[(aD_k + 1)\omega_k \Delta t\right] \end{aligned}\right\} \quad a = 0,1,2,3,\ldots$$

permits Eqs. (1) and (2) to be rewritten as

$$A_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{D_k}\left[\cos(\omega_k n\Delta t)\sum_{a=0}^{\beta_k} c\left[\Delta t\left(n + aD_k\right)\right]\right] \tag{12}$$

$$B_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{D_k}\left[\sin(\omega_k n\Delta t)\sum_{a=0}^{\beta_k} c\left[\Delta t\left(n + aD_k\right)\right]\right] \tag{13}$$

where $\beta_k = (N/D_k) - 1$. The identities

$$\sin(\theta - \pi) = -\sin\theta, \text{ and}$$

$$\cos(\theta - \pi) = \cos\theta$$

or

$$\sin\left[\omega_k \Delta t\left(n - \frac{D_k}{2}\right)\right] = -\sin\ (\omega_k n \Delta t)$$

$$\cos\left[\omega_k \Delta t\left(n - \frac{D_k}{2}\right)\right] = \cos\ (\omega_k n \Delta t)$$

permit Eqs. (12) and (13) to be written as

$$A_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{D_k/2}\left[\cos\ (\omega_k n\Delta t)\sum_{a=0}^{\beta_k}\left\{c\left[\Delta t(n + aD_k)\right] - c\left[\Delta t\left(n + \frac{D_k}{2} + aD_k\right)\right]\right\}\right] \tag{14}$$

$$B_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{D_k/2}\left[\sin\ (\omega_k n\Delta t)\sum_{a=0}^{B_k}\left\{c\left[\Delta t(n + aD_k)\right] - c\left[\Delta t\left(n + \frac{D_k}{2} + aD_k\right)\right]\right\}\right] \tag{15}$$

Finally, the identities sin $(-\theta) = -\sin\theta$, cos $(-\theta) = \cos\theta$, sin $(\pi/2)$ = cos (π) = 1, and sin (π) = cos $(\pi/2)$ = 0 permit Eqs. (14) and (15) to be written as

$$A_{ck} = \frac{\Delta t}{T}\sum_{n=1}^{\gamma_k}\left[\cos\ (\omega_k n\Delta t)\sum_{a=0}^{\beta_k}\left\{c\left[\Delta t(n + aD_k)\right] - c\left[\Delta t\left(n + \frac{D_k}{2} + aD_k\right)\right] - c\left[\Delta t\left(\frac{D_k}{2} - n + aD_k\right)\right] + c\left[\Delta t\left(\frac{D_k}{2} - n + \frac{D_k}{2} + aD_k\right)\right]\right\}\right]$$
$$- \frac{\Delta t}{T}\sum_{a=0}^{\beta_k}\left\{c\left[\Delta t\left(\frac{D_k}{2} + aD_k\right)\right] - c\left[\Delta t\left(\frac{D_k}{2} + \frac{D_k}{2} + aD_k\right)\right]\right\} \tag{16}$$

$$B_{ck} = \frac{\Delta t}{T}\left[\sum_{n=1}^{\gamma_k} \sin\ (\omega_k n\Delta t) \sum_{a=0}^{\beta_k} \left\{c\left[\Delta t(n + aD_k)\right] - c\left[\Delta t\left(n + \frac{D_k}{2} + aD_k\right)\right]\right.\right.$$

$$\left.\left. + c\left[\Delta t\left(\frac{D_k}{2} - n + aD_k\right)\right] - c\left[\Delta t\left(\frac{D_k}{2} - n + \frac{D_k}{2} + aD_k\right)\right]\right\}\right]$$

$$+ \frac{\Delta t}{T} \sum_{a=0}^{B_k} \left\{c\left[\Delta t\left(\frac{D_k}{4} + aD_k\right)\right] - c\left[\Delta t\left(\frac{D_k}{4} + \frac{D_k}{2} + aD_k\right)\right]\right\} \tag{17}$$

where $\gamma_k = (D_k/4) - 1$. Equations (16) and (17) represent the algorithm used in the hybrid program. The summation over "a" is performed between interrupts during the experiment and is called the "additive portion" of the FFT. At the end of the data-taking period, the summation over n (called the "multiplicative portion" of the FFT) and the calculation of the human operator's describing function [using Eqs. (8) and (9)], can be performed in less than three seconds.

The hybrid computer program is written in a Fortran IV language which includes hybrid commands. The program is listed in Appendix A. Table I lists the values of the experimental parameters, including those which characterize the system input. Figure 5 is a flow diagram of the additive portion of the FFT. Figure 6 shows a flow diagram of the hybrid program, and lists the time taken by each part of the program, both for the FFT version and for the version written the old way [directly computing Eqs. (1) through (4)]. As shown in Figure 6, the FFT permits a saving of nearly ten minutes per run, effectively reducing the run time to the time required to take the data and print the results.

RESULTS

An initial check of the hybrid program was made by taking measurements across known filters. The results shown in Figures 7 and 8 are quite accurate, and are repeatable.

Measurements were then taken of the author's tracking performance in a control loop, as shown in Figures 1 and 4. Ten runs were made with each of the controlled elements, 1/s and $1/s^2$. The describing functions shown in Figures 2 and 3 are comparable with established results (ref. 3).



JA(K) = (NO. OF SAMPLES PER CYCLE FOR ω_k)/2 • DIVISOR (K)
JB(K) = START OF STORAGE AREA FOR ω_k
JC(K) = DIVISOR (K)

Figure 5.- Flow chart for additive portion of FFT



	WITHOUT FFT	WITH FFT	
T_0	180	180	ONCE ONLY INITIALIZATION
T_1	24	24	EXPERIMENT RUNNING
T_2	216	216	
T_3	600	3	PROCESS DATA & PRINT DATA & RESET
T_4	5	5	
T_5	15	15	
	860 SECS	263 SECS	TOTAL (EXCEPT T_0)

Figure 6.- Flow diagram and times for computer runs



Figure 7.- Measurement of a known filter



Figure 8.- Measurement of a known filter

TABLE I

PARAMETER VALUES USED FOR THE HYBRID PROGRAM

k	A_k volts	ω_k rad/sec	k	A_k volts	ω_k rad/sec
1	.2	26.18	8	-1.	1.309
2	- .2	15.71	9	1.	.8727
3	.2	8.727	10	-1.	.5818
4	- .2	6.545	11	1.	.4363
5	.2	4.363	12	-1.	.2909
6	-1.	2.618	13	1.	.1745
7	1.	1.745	14	-1.	.1164

Δt = time between interrupts = .02 sec

T_1 = warm-up time before data-taking = 24 sec

T = period of data-taking = 216 sec

$$i(n\Delta t) = \text{system input} = \sum_{k=1}^{14} A_k \sin(\omega_k n\Delta t)$$

No comparison is made between results for the programs with and without the FFT (on Figures 2, 3, 7, 8) because the results are identical, as is shown analytically in the derivation of Eqs. (16) and (17). The comparison between the computation times for the two programs (Figure 6) however, indicates the substantial savings obtained by using the FFT. The only penalty paid for the reduced computational time is an increase in the complexity of the written Fortran program, as shown in Appendix A.

National Aeronautics and Space Administration
Electronics Research Center
Cambridge, Massachusetts, November 1968
125-19-01-11-25

REFERENCES

1. McRuer, D. T., et al.: New Approaches to Human-Pilot/Vehicle Dynamic Analysis. AFFDL-TR-67-150, Feb. 1968.

2. Elkind, J. I., et al.: An Optimal Control Method for Predicting Control Characteristics and Display Requirements for Manned-Vehicle Systems. AFFDL-TR-67-187, Apr. 1968.

3. McRuer, D. T., et al.: Human Pilot Dynamics in Compensatory Systems. AFFDL-TR-65-15, July 1965.

4. Taylor, L. W., Jr.: Discussion of Spectral Human Response Analysis. NASA-University Annual Conference on Manual Control, Feb. 1966.

National Aeronautics and Space Administration
Electronics Research Center
Cambridge, Massachusetts, December, 1968
125-19-01-11-25

APPENDIX A

PROGRAM LISTING

```
    11/13/68                            80/80 LIST                                    PAGE NO. 000001
*JOB RSHIRLEY
*DATE 1,NOVEMBER,1968
*TITLE    PROGRAM TO MEASURE THE DESCRIBING FUNCTION USING THE FFT
*ASSIGN 1=MT2A,2=MT1B,3=MT1A
*ASSIGN 5=CR1A, 6=LP1A
*ASSIGN 7=CP1A
*FORTRAN S,GO
C
C
C MAIN PROGRAM
C
C
C      SENSE SWITCH FORMAT
C      DO NOT SET SENSE SWITCHES DURING COMPILATION
C      SET=SWITCH IN, LIGHT ON
C      NOT SET=SWITCH OUT, LIGHT OFF
C
C       SWITCH NO.      SET                                 NOT SET
C       ******************************************************************************
C           1 *  DO NOT READ DATA CARDS                   * READ DATA CARDS AND CALC. INPUT
C           2 *  DO NOT CALC. INPUT                       * CALC. INPUT
C           3 *  DO NOT PROCESS DATA                      * PROCESS DATA
C           4 *  DO NOT USE UNIT 3 TO SAVE DATA           * USE UNIT 3 TO SAVE DATA
C           5 *  DO A DATA DUMP                           * NO DATA DUMP
C           6 * DO PUNCH OUT DATA                         * DO NOT PUNCH OUT DATA
C
C
C
C
C                              ************** HUMAN  **************        SYSTEM
C        INPUT        ERROR    *           *  OUTPUT *             *     OUTPUT
C        (PUT)    + ***  (E)   *   HUMAN    *  (ONE)  *   VEHICLE   *      (TWO)
C       ********** ***********             ***********            **************
C                    ***       *  OPERATOR  *          *  DYNAMICS  *      *
C                     * -      *            *          *            *      *
C                     *        **************          **************      *
C                     *                                                    *
C                     *                                                    *
C                     ******************************************************
C
C
C FORMAT STATEMENTS
  100 FORMAT(1H1,3X,1HK,6X,8HFREQ.(K),7X,5HAE(K),10X,5HBE(K),10X,
     15HA1(K),10X,5HB1(K),10X,5HA2(K),10X,5HB2(K),//,(I5,7E15.4))
  101 FORMAT(///,4X,1HK,6X,8HFREQ.(K),7X,8HAREM1(K),7X,8HBREM1(K),7X,
     18HAREM2(K),7X,8HBREM2(K),//,(I5,5E15.4))
  102 FORMAT(1E15.8)
  103 FORMAT(///,//,4X,1HK,6X,8HFREQ.(K),7X,6HAR1(K),9X,7HPHA1(K),8X,
     16HAR2(K),9X,7HPHA2(K),//,(I5,5E15.4))
  104 FORMAT (1H1,4X,1HK,10X,4HE(K),9X,6HONE(K),9X,6HTWO(K),//,(I5,3E15.
     24))
  105 FORMAT (1H1,4X,1HK,9X,6HPUT(K),9X,4HE(K),11X,6HONE(K),9X,6HTWO(K)
     1,//,(I5,4E15.4))
  106 FORMAT (1H1,4X,1HK,9X,6HPUT(K),//,(I5,1E15.4))
  107 FORMAT (///,10X,8HPUTSQ = ,1E15.4,/,10X,8HERRSQ = ,1E15.4,10X,
     114HERRSQ/PUTSQ = ,1E15.4,/,10X,8HONESQ = ,1E15.4, 10X,14HONESQ/PUT
     2SQ = ,1E15.4,/,10X,8HTWOSQ = ,1E15.4,10X,14HTWOSQ/PUTSQ = ,1E15.4)
```

```
  108 FORMAT(///,4X,1HK,6X,8HFREQ.(K),7X,7HREM1(K),8X,7HREM2(K),//,
     1(I5,3E15.4))
  109 FORMAT (10X,3I10)
  110 FORMAT (//,4X,5HJA(K),4X,5HJB(K),4X,5HJC(K),//,(2X,I5,4X,I5,4X,I5)
     1)
  111 FORMAT (1H1,/,4X,1HK,5X,4HW(K),//,(I5,E15.4))
  112 FORMAT (///,4X,23HREMNANT POWER AT ONE = ,1E15.4,/,4X,
     123HREMNANT POWER AT TWO = ,1E15.4)
  113 FORMAT (///,4X,1HK,6X,8HFREQ.(K),7X,8HREMA1(K),7X,8HREMA2(K),//,
     1(I5,3E15.4))
  114 FORMAT (///,10X,$THE FOLLOWING REMNANT VALUES ARE CORRECTED FOR TH
     1E LOOP GAINS,/,$I.E., PHI ACTUAL = PHI MEAS. TIMES 1 + YPYC SQUARE
     2D$)
  115 FORMAT (6E10.4)
  116 FORMAT (/,4X,8HCONS2 = ,1E15.4)
  117 FORMAT (//,4X,7HX1SQ = ,1E15.4,10X,7HX2SQ = ,1E15.4)
C
C DIMENSION STATEMENTS
      COMMON PUT(1080),E(1080),ONE(1080),TWO(1080),KONK,LOP,KANK,
     1N,M,MO,KRINK,PUTSQ,ERRSQ,ONESQ,TWOSQ
      COMMON JA(29),JB(29),JC(29),WS(4300),KL(29),KLE(29)
      DIMENSION W(30),AE(15),BE(15),A1(15),B1(15),A2(15),B2(15),
     1AREM1(15),BREM1(15),AREM2(15),BREM2(15)
      DIMENSION AR1(15),AR2(15),PHA1(15),PHA2(15),REM1(15),REM2(15)
      DIMENSION AM(15)
      DIMENSION REMA1(15),REMA2(15)
C
C
C     THE PROGRAM CAN BE RECALLED TO THIS POINT BY IFINITIA AT ANY TIME
C     BY HITTING INTERRUPT 33 AND TYPING A CARRIAGE RETURN
      CALL IFINITIA
C
C     CONNECT THE CONSOLE
S     EOM 031120
C
C     PUT THE ANALOG COMPUTER INTO IC MODE
      CALL IC
C
C     SENSE SWITCH4 DETERMINES IF REWIND 3 OR NOT
      IF (SENSE SWITCH4) 98,99
   99 REWIND 3
   98 CONTINUE
C
C     SENSE SWITCH1 DETERMINES IF FREQUENCY AND REGISTER VALUE CARDS ARE
C     READ
      IF (SENSE SWITCH1) 23,24
C
C     SENSE SWITCH2 DETERMINES IF INPUT CALCULATED
   23 IF (SENSE SWITCH2) 17,26
C
C     READ IN 29 FREQUENCIES IN ORDER, HIGH FREQUENCIES FIRST,
C     IN RADIANS/SECOND, STARTING WITH A REMNANT FREQUENCY AND ALTERNA-
C     TING THEREAFTER WITH THE INPUT FREQUENCIES.  THE FREQUENCIES MUST
C     ALL BE INTEGER MULTIPLES OF 2(PI)/RUNTIME, AND THE NUMBER OF
C     SECONDS PER CYCLE MUST EQUAL 4I(DELTA), WHERE I IS AN INTEGER AND
C     DELTA IS THE TIME BETWEEN DATA POINTS.
   24 READ(5,102) (W(K),K=1,29)
```

```
      WRITE(6,111) (K,W(K),K=1,29)
C
C     READ IN THE REGISTER VALUES FOR THE ADDITIVE PART OF THE FFT
      READ(5,109) (JA(K),JB(K),JC(K),K=1,29)
      WRITE(6,110) (JA(K),JB(K),JC(K),K=1,29)
C
C
C        INPUT (CALCULATED EVERY DELT)
C     CONS1 SCALES THE INPUT
   26 CONS1=9.5
C     REWIND INPUT TAPE PRIOR TO STORING INPUT
      REWIND 1
C     ONW SETS THE STARTING TIME FOR THE INPUT, AND IS NEGATIVE
C       SO THAT THE HUMAN OPERATOR IS IN A STEADY-STATE TRACKING
C       CONDITION WHEN THE ONSET OF DATA-TAKING OCCURS
      ONW=-1080.
C     DELT IS THE TIME INCREMENT BETWEEN INPUT VALUES, AND MUST EQUAL
C     DELTA, THE TIME INCREMENT BETWEEN DATA POINTS.
      DELT=.02
C     THE AM(K) SCALE THE INPUT SINUSOIDS
      AM(1)=.2
      AM(2)=-.2
      AM(3)=.2
      AM(4)=-.2
      AM(5)=.2
      AM(6)=-1.
      AM(7)=1.
      AM(8)=-1.
      AM(9)=1.
      AM(10)=-1.
      AM(11)=1.
      AM(12)=-1.
      AM(13)=1.
      AM(14)=-1.
      AM(15)=1.
C
C     THE INPUT IS CALCULATED IN BLOCKS OF 540 VALUES AND STORED ON
C       MAGNETIC TAPE (UNIT 1).
      DO 19 J=1,30
      DO 1 K=1,540
      PUT(K)=0.
      ONW=ONW+1.
      T=ONW*DELT
      DO 20 L=1,14
      PUT(K)=PUT(K)+AM(L)*SIN(T*W(2*L))
   20 CONTINUE
      PUT(K)=CONS1*PUT(K)
    1 CONTINUE
      M=1
      N=540
      CALL BUFFEROUT(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
   19 CONTINUE
C
C
   17 CONTINUE
C     THE INTERRUPT IS CONNECTED, BUT NOT ENABLED
```

```
      CONNECT (40,INTR)
      EOM 020020
      POT =00700000


      INITIALIZE FOR THE RUN

      ZERO THE BUFFER AREAS
      DO 92 K=1,1080
      PUT(K)=0.
      E(K)=0.
      ONE(K)=0.
      TWO(K)=0.
   92 CONTINUE

      ZERO THE DATA TAPE
      REWIND 2
      DO 91 K=1,90
      N=540
      M=1
      CALL BUFFEROUT(2,1,E(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
   91 CONTINUE

      THE MAGNETIC TAPE UNITS ARE INITIALIZED, UNIT 1 FOR THE INPUT,
        AND UNIT 2 FOR THE DATA.
      REWIND 1
      REWIND 2

      ZERO THE REGISTERS WHERE THE FOURIER COEFFICIENTS ARE TO
        BE CALCULATED.
      DO 31 K=1,15
      AE(K)=0.
      BE(K)=0.
      A1(K)=0.
      B1(K)=0.
      A2(K)=0.
      B2(K)=0.
      AREM1(K)=0.
      BREM1(K)=0.
      AREM2(K)=0.
      BREM2(K)=0.
   31 CONTINUE

      INITIALLIZE FOR THE ADDITIVE PART OF THE FFT
      DO 10 NOW=1,29
      KL(NOW)=-1
   10 CONTINUE
      DO 14 NOW=1,4300
      WS(NOW)=0.
   14 CONTINUE

INITIALLIZE THE COUNTERS FOR THE RUN
        KRINK IS A COUNTER TO DETERMINE THE LOCATION FROM WHICH THE
          NEXT INPUT VALUE SHOULD BE TAKEN FROM
      KRINK=0
        KONK COUNTS THE INTERRUPTS, DETERMINES WHEN THE ONSET OF DATA-
```

```
C        TAKING SHOULD OCCUR, AND WHEN DATA-TAKING IS COMPLETED.
      KONK=0
C     KANK IS THE HALF REGISTER COUNTER, 1 TO 540
      KANK=0
C       MO DETERMINES WHICH HALF OD THE INPUT BUFFER IS BEING USED
      MO=0
C       B IS A COUNTER ON THE DATA USED DURING THE DATA-PROCESSING
      B=-1.
C       LOP IS THE FLAG SET BY INTR TO END DATA-TAKING
      LOP=0
C     PUTSQ AND ERRSQ ARE THE INTEGRAL SQUARE INPUT AND ERROR
      PUTSQ=0.
      ERRSQ=0.
      ONESQ=0.
      TWOSQ=0.
      X1SQ=0.
      X2SQ=0.
C
C     INITIALLIZE INPUT BUFFER FOR THE RUN, I.E., FILL BOTH HALVES
C       WITH INPUT VALUES
      M=1
      N=540
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      M=541
      N=1080
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
C
C     WAIT TO START RUN ON SIGNAL FROM THE OPERATOR
      PAUSE
C
C
  511 CONTINUE
S     SKS 030000
S     BRU 513S
/     GO TO 511
  513 CONTINUE
C     PUT THE ANALOG COMPUTER INTO COMPUTE MODE
      CALL COMPUTE
C
C     ENABLE THE INTERRUPT
S     EOM 031032
C
C
   11 CONTINUE
C     CHECK TO SEE IF IT IS THE END OF DATA-TAKING
      IF (LOP.EQ.1) GO TO 2
C     IT IS NOT THE END OF DATA TAKING, WAIT FOR INTERRUPT
      GO TO 11
C
C
C     IT IS THE END OF DATA TAKING, GO ON
    2 CONTINUE
C     TURN OFF THE INTERRUPT
S     EOM 031033
C     PUT THE ANALOG COMPUTER INTO THE HOLD MODE
```

```
      CALL HOLD
C     TAKE THE INTEGRAL SQUARE MEASURES
      CALL ADL(4,PUTSQ,ERRSQ,ONESQ,TWOSQ,X1SQ,X2SQ)
C     PUT THE ANALOG COMPUTER INTO IC MODE
      CALL IC
C
C
C     SENSE SWITCH3 DETERMINES WHETHER TO PROCESS THE DATA AND TYPE
C       THE RESULTS, OR WHETHER TO RE-INITIALLIZE FOR THE NEXT RUN
      IF (SENSE SWITCH3) 17,22
C
C     DATA PROCESSING IS DESIRED, GO ON
   22 CONTINUE
C     THE FOLLOWING SAVES THE LAST 540 DATA POINTS
      N=1080
      M=541
      CALL BUFFEROUT(2,1,E(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFEROUT(2,1,ONE(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFEROUT(2,1,TWO(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
C
C
C     THE FOLLOWING PERMITS A TOTAL OR PARTIAL TAPE DUMP
      IF (SENSE SWITCH5) 89,90
   89 REWIND 1
      REWIND 2
      N=540
      M=1
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      N=1080
      M=541
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      WRITE(6,106) (J,PUT(J),J=1,1080)
C     GO PAST THE INITIAL SPURIOUS DATA POINTS (CAUSED BY INTERRUPT
C     ROUTINE).
      M=1
      N=540
      DO 16 J=1,3
      CALL BUFFERIN(2,1,E(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
   16 CONTINUE
C     NO DETERMINES EXTENT OF THE DUMP
      NO=3
      DO 88 K=1,NO
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(2,1,E(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(2,1,ONE(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(2,1,TWO(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      L=(K-1)*540
```

```
      WRITE(6,105) ((L+J),PUT(J),E(J),ONE(J),TWO(J),J=1,540)
   88 CONTINUE
   90 CONTINUE
C
C
C     PROCESS THE DATA FOR 10,800 POINTS
C
C
C     MULTIPLICATIVE PART OF THE FFT USED TO CALCULATE THE FOURIER
C        COEFFICIENTS
C
C     CALCULATE THE FOURIER COEFFICIENTS FOR THE REMNANT
      DEZ=3.1415927
      DO 21 K=1,29,2
      NE=(K+1)/2
      LEB1=JB(K)+JA(K)-1
      LEE1=JB(K)+2*JA(K)-1
      LEB2=JB(K)+2*JA(K)-1
      LEE2=JB(K)+3*JA(K)-1
      LEM1=LEE1-JA(K)/2
      LEM2=LEE2-JA(K)/2
      AREM1(NE)=0.
      AREM2(NE)=0.
      BREM1(NE)=0.
      BREM2(NE)=0.
      JST=(JA(K)/2)-1
      DO 27 JI=1,JST
      UOU=FLOAT(JI)
      VOU=FLOAT(JA(K))
      TOU=UOU/VOU
      SINUS=SIN(DEZ*TOU)
      COSUS=SQRT(1.-SINUS**2.)
      AREM1(NE)=AREM1(NE)+(WS(LEB1+JI)+WS(LEE1-JI))*SINUS
      AREM2(NE)=AREM2(NE)+(WS(LEB2+JI)+WS(LEE2-JI))*SINUS
      BREM1(NE)=BREM1(NE)+(WS(LEB1+JI)-WS(LEE1-JI))*COSUS
      BREM2(NE)=BREM2(NE)+(WS(LEB2+JI)-WS(LEE2-JI))*COSUS
   27 CONTINUE
      AREM1(NE)=AREM1(NE)+WS(LEM1)
      AREM2(NE)=AREM2(NE)+WS(LEM2)
      BREM1(NE)=BREM1(NE)-WS(LEE1)
      BREM2(NE)=BREM2(NE)-WS(LEE2)
   21 CONTINUE
C
C     CALCULATE THE FOURIER COEFFICIENTS FOR THE DESCRIBING FUNCTION
      DO 28 K=2,28,2
      NE=K/2
      LEBE=JB(K)-1
      LEEE=JB(K)+JA(K)-1
      LEB1=JB(K)+JA(K)-1
      LEE1=JB(K)+2*JA(K)-1
      LEB2=JB(K)+2*JA(K)-1
      LEE2=JB(K)+3*JA(K)-1
      LEME=LEEE-JA(K)/2
      LEM1=LEE1-JA(K)/2
      LEM2=LEE2-JA(K)/2
      AE(NE)=0.
      BE(NE)=0.
```

```
      A1(NE)=0.
      B1(NE)=0.
      A2(NE)=0.
      B2(NE)=0.
      JST=(JA(K)/2)-1
      DO 29 JI=1,JST
      UOU=FLOAT(JI)
      VOU=FLOAT(JA(K))
      TOU=UOU/VOU
      SINUS=SIN(DEZ*TOU)
      COSUS=SQRT(1.-SINUS**2.)
      AE(NE)=AE(NE)+(WS(LEBE+JI)+WS(LEEE-JI))*SINUS
      A1(NE)=A1(NE)+(WS(LEB1+JI)+WS(LEE1-JI))*SINUS
      A2(NE)=A2(NE)+(WS(LEB2+JI)+WS(LEE2-JI))*SINUS
      BE(NE)=BE(NE)+(WS(LEBE+JI)-WS(LEEE-JI))*COSUS
      B1(NE)=B1(NE)+(WS(LEB1+JI)-WS(LEE1-JI))*COSUS
      B2(NE)=B2(NE)+(WS(LEB2+JI)-WS(LEE2-JI))*COSUS
   29 CONTINUE
      AE(NE)=AE(NE)+WS(LEME)
      A1(NE)=A1(NE)+WS(LEM1)
      A2(NE)=A2(NE)+WS(LEM2)
      BE(NE)=BE(NE)-WS(LEEE)
      B1(NE)=B1(NE)-WS(LEE1)
      B2(NE)=B2(NE)-WS(LEE2)
   28 CONTINUE
C
C
C     CALCULATE THE HUMAN OPERATORS DESCRIBING FUNCTION AND REMNANT
C
      DO 32 K=1,14
      DENOM=AE(K)*AE(K)+BE(K)*BE(K)
      AR1(K)=SQRT((A1(K)*A1(K)+B1(K)*B1(K))/DENOM)
      AR2(K)=SQRT((A2(K)*A2(K)+B2(K)*B2(K))/DENOM)
      PHA1(K)=57.3*(ATAN2(B1(K),A1(K))-ATAN2(BE(K),AE(K)))
      PHA2(K)=57.3*(ATAN2(B2(K),A2(K))-ATAN2(BE(K),AE(K)))
C     USE THE ASSUMPTION THAT THE HUMANS PHASE LEAD IS LESS THAN
C        180 DEGRESS TO CORRECT FOR THE LOSS OF PHASE INFORMATION
C        IN THE ARC-TANGENT ROUTINES
      IF (PHA1(K).LT.180.) GO TO 93
      PHA1(K)=PHA1(K)-360.
   93 IF (PHA2(K).LT.180.) GO TO 94
      PHA2(K)=PHA2(K)-360.
   94 CONTINUE
   32 CONTINUE
C
C     CONS2 SCALES THE REMNANT
C     CONS2=(DELT)**2/4(PI)T
C     DELT=.02, T=216
      CONS2=1.47E-7
      DO 33 K=1,15
      REM1(K)=CONS2*(AREM1(K)*AREM1(K)+BREM1(K)*BREM1(K))
      REM2(K)=CONS2*(AREM2(K)*AREM2(K)+BREM2(K)*BREM2(K))
   33 CONTINUE
C     INTERPOLATE FOR THE REMNANT AT THE INPUT FREQUENCIES
      DO 230 K=1,14
      REMA1(K)=REM1(K)+(REM1(K+1)-REM1(K))*(W(2*K)-W(2*K-1))/(W(2*K+1)-
     1W(2*K-1))
```

```
      REMA2(K)=REM2(K)+(REM2(K+1)-REM2(K))*(W(2*K)-W(2*K-1))/(W(2*K+1)-
     1W(2*K-1))
  230 CONTINUE
C     CALCULATE THE REMNANT POWER
      REMPW1=0.
      REMPW2=0.
      DO 231 K=1,14
      REMPW1=REMPW1+REMA1(K)*(W(2*K-1)-W(2*K+1))
      REMPW2=REMPW2+REMA2(K)*(W(2*K-1)-W(2*K+1))
  231 CONTINUE
C
C
C     WRITE OUT THE FOURIER COEFFICIENTS OF THE SYSTEM ERROR (E), THE
C        HUMANS OUTPUT (ONE), AND OF THE SYSTEM OUTPUT (TWO).
      WRITE(6,100) (K,W(2*K),AE(K),BE(K),A1(K),B1(K),A2(K),B2(K),K=1,14)
C
C     WRITE OUT THE FOURIER COEFFICIENTS OF THE REMNANT AT THE HUMANS
C        OUTPUT (ONE), AND AT THE SYSTEM OUTPUT (TWO).
      WRITE(6,101) (K,W(2*K-1),AREM1(K),BREM1(K),AREM2(K),BREM2(K),K=1,1
     15)
C
C     WRITE OUT THE HUMAN OPERATORS DESCRIBING FUNCTION AND REMNANT AS
C     WELL AS THE SYSTEM OPEN LOOP DESCRIBING FUNCTION AND REMNANT
      WRITE(6,108) (K,W(2*K-1),REM1(K),REM2(K),K=1,15)
      WRITE(6,116) CONS2
C     WRITE OUT THE HUMANS REMNANT AT THE INPUT FREQUENCIES
      WRITE (6,113) (K,W(2*K),REMA1(K),REMA2(K),K=1,14)
C
      WRITE(6,103) (K,W(2*K),AR1(K),PHA1(K),AR2(K),PHA2(K),K=1,14)
      DO 239 K=1,14
      REMA1(K)=REMA1(K)*((1.+AR2(K))**2.)
      REMA2(K)=REMA2(K)*((1.+AR2(K))**2.)*(AR1(K)**2.)/(AR2(K)**2.)
  239 CONTINUE
      WRITE(6,114)
      WRITE(6,113) (K,W(2*K),REMA1(K),REMA2(K),K=1,14)
      WRITE(6,116) CONS2
C     WRITE OUT THE ERROR SCORES
      ERRSY=ERRSQ/PUTSQ
      ONESY=ONESQ/PUTSQ
      TWOSY=TWOSQ/PUTSQ
      WRITE(6,107) (PUTSQ,ERRSQ,ERRSY,ONESQ,ONESY,TWOSQ,TWOSY)
      WRITE(6,117) X1SQ,X2SQ
      PUTSQ=PUTSQ/216.
      ERRSQ=ERRSQ/216.
      ONESQ=ONESQ/216.
      TWOSQ=TWOSQ/216.
      WRITE(6,107) (PUTSQ,ERRSQ,ERRSY,ONESQ,ONESY,TWOSQ,TWOSY)
C     WRITE OUT THE REMNANT POWER
      WRITE(6,112) (REMPW1,REMPW2)
      REMPW1=0.
      REMPW2=0.
      DO 240 K=1,14
      REMPW1=REMPW1+REMA1(K)*(W(2*K-1)-W(2*K+1))
      REMPW2=REMPW2+REMA2(K)*(W(2*K-1)-W(2*K+1))
  240 CONTINUE
      WRITE(6,112) (REMPW1,REMPW2)
C     SENSE SWITCH 6 DETERMINES WHETHER TO PUNCH OUT THE ANSWERS ON CARDS
```

```
      IF (SENSE SWITCH6) 35,36
C     PUNCH OUT THE DATA ON CARDS, 14 CARDS WITH AR1,PHA1,AR2,PHA2,
C         REMA1, AND REMA2, PLUS 1 CARD WITH PUTSQ,ERRSQ,ONESQ,TWOSQ,
C         REMPW1, AND REMPW2.
   35 WRITE(7,115) (AR1(K),PHA1(K),AR2(K),PHA2(K),REMA1(K),REMA2(K),K=1,
     114)
      WRITE(7,115) (PUTSQ,ERRSQ,ONESQ,TWOSQ,REMPW1,REMPW2)
   36 CONTINUE
C
C
C     THE FOLLOWING STORES EITHER ONE(M) OR TWO(M) ON UNIT 3 FOR
C        LATER PROCESSING FOR THE REMNANT
C     SENSE SWITCH4 DETERMINES WHETHER TO SAVE ONE(M) OR TWO(M)
      IF (SENSE SWITCH4) 95,97
   97 REWIND 2
      REWIND 1
      M=1
      N=540
C     THE FOLLOWING AVOIDS THE SPURIOUS DATA POINTS
      DO 521 K=1,3
      CALL BUFFERIN(2,1,TWO(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
  521 CONTINUE
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      DO 96 K=1,20
      CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(2,1,E(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(2,1,ONE(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFERIN(2,1,TWO(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFEROUT(3,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFEROUT(3,1,E(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFEROUT(3,1,ONE(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
      CALL BUFFEROUT(3,1,TWO(M),2*(N-M+1),ISTATUS)
      CALL GOTO(ISTATUS)
   96 CONTINUE
   95 CONTINUE
C
C
C     RETURN TO INITIALLIZE FOR THE NEXT RUN
      GO TO 17
  200 STOP
C
C
C     INTERRUPT SUBROUTINE (INTERNAL)
C     INTR SERVICES THE INTERRUPT
      SUBROUTINE INTR
C
```

```
C     KRINK IS THE REGISTER COUNTER, 1 TO 1080
      KRINK=KRINK+1
C     KONK IS THE TOTAL COUNTER, 1 ON UP
      KONK=KONK+1
C     KANK IS THE HALF REGISTER COUNTER, 1 TO 500
      KANK=KANK+1
      IF (KONK.GT.1080) GO TO 1
    2 IF (KANK.GE.541) GO TO 3
    4 CALL DAL(0,PUT(KRINK))
   12 RETURN
    1 IF (KONK.GT.11880) GO TO 5
      IF (KANK.GE.541) GO TO 6
    8 CALL DAL(0,PUT(KRINK))
      CALL ADL(0,E(KRINK),ONE(KRINK),TWO(KRINK))
C     ADDITIVE PART OF THE FAST FOURIER TRANSFORM
C
C     KL= A COUNTER, 0 TO (2*JA(K)*JC(K)-1)
C     JA=HALF THE NUMBER OF CYCLES PER SECOND/DIVISOR
C     JB=REGISTER START FOR EACH FREQUENCY
C     JC=DIVISOR USED ON HALF THE NUMBER OF SAMPLES PER CYCLE TO GET JA
C
C     ADDITIVE PART OF THE FFT
C
C     ADD THE DATA INTO THE WS(K)
      DO 27 K=1,29
   23 KL(K)=KL(K)+1
      JNK=JA(K)*JC(K)
      IF (KL(K).GE.JNK) GO TO 32
      KLE(K)=KL(K)/JC(K) + JB(K)
      GO TO 24
   32 JAK=2*JA(K)*JC(K)
      IF(KL(K).GE.JAK) GO TO 33
      KLE(K)=(KL(K)/JC(K))-JA(K)+JB(K)
      GO TO 25
   33 KL(K)=-1
      GO TO 23
   24 JD=KLE(K)
      JE=KLE(K)+JA(K)
      JF=KLE(K)+2*JA(K)
      WS(JD)=WS(JD)+E(KRINK)
      WS(JE)=WS(JE)+ONE(KRINK)
      WS(JF)=WS(JF)+TWO(KRINK)
      GO TO 26
   25 JD=KLE(K)
      JE=KLE(K)+JA(K)
      JF=KLE(K)+2*JA(K)
      WS(JD)=WS(JD)-E(KRINK)
      WS(JE)=WS(JE)-ONE(KRINK)
      WS(JF)=WS(JF)-TWO(KRINK)
   26 CONTINUE
   27 CONTINUE
      RETURN
    3 KANK=1
      IF (MO.GE.1) GO TO 7
      MO=1
      N=540
      M=1
```

```
    9 CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      GO TO 4
    7 MO=0
      N=1080
      M=541
      KRINK=1
      GO TO 9
    6 KANK=1
      IF (MO.GE.1) GO TO 10
      MO=1
      N=540
      M=1
   11 CALL BUFFERIN(1,1,PUT(M),2*(N-M+1),ISTATUS)
      CALL BUFFEROUT(2,1,E(M),2*(N-M+1),ISTATUS)
      CALL BUFFEROUT(2,1,ONE(M),2*(N-M+1),ISTATUS)
      CALL BUFFEROUT(2,1,TWO(M),2*(N-M+1),ISTATUS)
      GO TO 8
   10 MO=0
      N=1080
      M=541
      KRINK=1
      GO TO 11
    5 LOP=1
      GO TO 12
C
C
      SUBROUTINE GOTO (ISTATUS)
C     SUBROUTINE (INTERNAL) TO HANDLE TAPE READ AND WRITES
    7 GO TO (6,4,5,5,5)ISTATUS
    6 GO TO 7
    5 WRITE (102,200)ISTATUS
    4 RETURN
  200 FORMAT ($BUFFERIN  STATUS WORD =$,I2)
      END
*LOAD X
*DATA
39.269908
26.179938
19.634954
15.707963
13.089969
8.7266460
7.8539816
6.5449847
5.2359877
4.3633230
3.1415927
2.6179939
2.1816615
1.7453292
1.5707963
1.3089970
1.0471975
.87266462
.72722052
.58177642
.52359877
```

.43633231
.34906585
.29088821
.26179938
.17453292
.14544410
.11635528
.08726646

4	1	1
6	13	1
8	31	1
10	55	1
12	85	1
18	121	1
20	175	1
24	235	1
30	307	1
36	397	1
50	505	1
60	655	1
72	835	1
90	1051	1
50	1321	2
50	1651	3
60	1471	2
60	1801	3
54	1981	4
54	2143	5
50	2305	6
60	2455	6
50	2635	9
60	2785	9
60	2965	10
108	3415	10
90	3145	10
90	3739	15
90	4009	20

*FIN

APPLICATION OF A MODIFIED FAST FOURIER TRANSFORM TO CALCULATE HUMAN OPERATOR DESCRIBING FUNCTION

By Richard S. Shirley
Electronics Research Center

ABSTRACT

A modified fast Fourier transform (FFT) is used in a hybrid computer program to permit processing of tracking data during a run to yield the human operator's describing function almost immediately after the data-taking period. The computer processing time is substantially reduced at no cost in accuracy.

STAR Category 08

NASA-Langley, 1969 —— 5 C-58

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Notes, and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION DIVISION

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546